

UnitLib

**The library of normalized unit groups of
modular group algebras**

Version 5.0.0

12 June 2025

**Olexandr Konovalov
Olena Yakimenko
Kamil Zabielski**

Olexandr Konovalov Email: obk1@st-andrews.ac.uk
Homepage: <https://olexandr-konovalov.github.io/>
Address: School of Computer Science
University of St Andrews
Jack Cole Building, North Haugh,
St Andrews, Fife, KY16 9SX, Scotland

Olena Yakimenko
Address: Department of Mathematics
Zaporizhzhia National University
Zaporizhzhia, Ukraine

Kamil Zabielski Email: kamil@zabielscy.com
Homepage: <https://limakzi.me/>
Address: Faculty of Computer Science
Białystok University of Technology
Białystok, Poland

Abstract

The UnitLib package extends the LAGUNA package and provides the library of normalized unit groups of modular group algebras of all finite p -groups of order up to 243 over the field of p elements.

It also contains a parallel implementation of the computation of the normalized unit group of a modular group algebra of a finite p -group (which should be retrieved from the GAP Small Groups Library) over the field of p elements.

Copyright

© 2006–2025 by Olexandr Kononov and Olena Yakimenko

UnitLib is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version. For details, see the FSF's own site <https://www.gnu.org/licenses/gpl.html>.

If you obtained UnitLib, we would be grateful for a short notification sent to one of the authors.

If you publish a result which was partially obtained with the usage of UnitLib, please cite it in the following form:

O. Kononov and O. Yakimenko. *UnitLib --- The library of normalized unit groups of modular group algebras*, Version 5.0.0, 2025 (<https://gap-packages.github.io/unitlib/>).

Acknowledgements

The first version of the package (as well as the subsequent version 2.1) was released in 2006, when the first author was a postdoctoral research collaborator at the Vrije Universiteit Brussels, Belgium. It is a pleasure of the first author to express his gratitude to Prof. Dr. Eric Jespers for his warm hospitality and to acknowledge the support provided by the Francqui Stichting grant ADSI107.

Both authors are very grateful to the members of the GAP team: Thomas Breuer, Stefan Kohl and Frank Lübeck for helpful suggestions. We would like to acknowledge Bettina Eick for communicating the package, and the referee for testing UnitLib and useful comments. Finally, we wish to thank the Centre for Interdisciplinary Research in Computational Algebra of the University of St Andrews and the Computational Cluster of the Kyiv National Taras Shevchenko University for offering their computer facilities for computations leading to the first version of this package, and to the Białystok University of Technology cluster used to produce the data for groups of order 243.

Contents

1	Introduction	4
1.1	General aims	4
1.2	Theoretical background	4
1.3	Installation and system requirements	5
2	UnitLib functions	6
2.1	MainFunctions	6
3	Implementation Details	10
3.1	Saving the data	10
3.2	Reading the data	11
4	An example of UnitLib usage	12
	References	14
	Index	15

Chapter 1

Introduction

1.1 General aims

Let KG be a group algebra of a finite p -group G over the field K of characteristic p , and let $V(KG)$ be the normalized unit group of KG . The pc-presentation of the group $V(KG)$ can be computed using the GAP package LAGUNA (<https://gap-packages.github.io/laguna/>), but for groups of orders 64 and more such computation will already take a lot of time.

The UnitLib package is an extension of the LAGUNA package that is focused on this problem. It contains the library of normalized unit groups of modular group algebras of finite p -groups over the field of p elements. This allows the user to retrieve the pre-computed group from the library instead of the time-consuming computation. The group created with UnitLib will have the same properties and attributes as the one computed with LAGUNA.

The version UnitLib 3.0.0 released in May 2009 also contained a parallel implementation of the computation of the normalized unit group of a modular group algebra of a finite p -group over the field of p elements, which works for groups from the GAP small groups library. It is developed on the base of the sequential version of this algorithm (which works for any p -group with no limitations) from the LAGUNA package. Parallelisation is implemented using the SCSCP package that is capable of connecting several local or remote GAP instances using the SCSCP protocol.

In April 2012, UnitLib 3.1.0 was updated to comply with GAP 4.5.

The current version of UnitLib provides the library of normalized unit groups $V(KG)$ for all p -groups of order up to 243.

If you need to work with groups of bigger orders, please write to the maintainers, because they may be already computed or we can compute them for you.

1.2 Theoretical background

Since the UnitLib package is an extension of the LAGUNA package [BKRS], we refer to the **LAGUNA: LAGUNA package** manual for the theoretical background. In particular, Chapter 3 (The basic theory behind LAGUNA) of that manual contains definitions of the modular group algebra and its normalized unit group, the power-commutator presentation of the group, and also more details about the algorithm for the computation of the pc-presentation of the normalized unit group of a modular group algebra of a finite p -group.

1.3 Installation and system requirements

UnitLib 5.0.0 requires at least GAP 4.10. The libraries of normalized unit groups of groups of orders less than 243 are included in the distribution. The data for order 243 is available as an optional download.

Because the UnitLib is an extension of the LAGUNA package, you must have the LAGUNA package installed. You can obtain it from the GAP homepage or from its homepage <https://gap-packages.github.io/laguna/>.

To use the UnitLib online help it is necessary to install the GAP4 package GAPDoc by Frank Lübeck and Max Neunhöffer, which is available from the GAP homepage or from <http://www.math.rwth-aachen.de/~Frank.Luebeck/GAPDoc/>.

UnitLib is distributed in standard formats (tar.gz, tar.bz2, .zip, -win.zip) and can be obtained from the GAP homepage or from <https://gap-packages.github.io/unitlib/>. To install UnitLib, unpack its archive into the pkg subdirectory of your GAP installation. When you don't have access to the directory of your main GAP installation, you can also install the package *outside the GAP main directory* by unpacking it inside a directory MYGAPDIR/pkg. Then to be able to load UnitLib you need to call GAP with the -l ";"MYGAPDIR" option.

Chapter 2

UnitLib functions

Since the main purpose of UnitLib is the data storage, it has only two main user functions to read the description of $V(KG)$ for the given catalogue number of G in the Small Groups Library of the GAP system, and to save the description of $V(KG)$ if the user would like to store it for the further usage for the group which is not contained in the library.

To use the UnitLib package first you need to load it as follows:

Example

```
gap> LoadPackage("unitlib");
-----
Loading UnitLib 5.0.0 (The library of normalized unit groups of modular group algebras)
by Olexandr Konovalov (https://olexandr-konovalov.github.io/) and
  Olena Yakimenko, and
  Kamil Zabielski (https://limakzi.me).
maintained by:
  Olexandr Konovalov (https://olexandr-konovalov.github.io/) and
  Kamil Zabielski (https://limakzi.me).
Homepage: https://gap-packages.github.io/unitlib
Report issues at https://github.com/gap-packages/unitlib/issues
-----
true
```

Examples below contain some functions from the LAGUNA package [BKRS], see their description in the LAGUNA manual (**LAGUNA: LAGUNA package**).

2.1 MainFunctions

2.1.1 PcNormalizedUnitGroupSmallGroup

▷ PcNormalizedUnitGroupSmallGroup(s, n) (function)

Returns: PcGroup

Let s be a power of prime p and n is an integer from $[1 \dots \text{NrSmallGroups}(s)]$. Then PcNormalizedUnitGroupSmallGroup(s, n) returns the normalized unit group $V(KG)$ of the modular group algebra KG , where G is SmallGroup(s, n) (see SmallGroup (**smallgrp: SmallGroup for group order and index**)) and K is a field of p elements.

Example

```
gap> PcNormalizedUnitGroupSmallGroup(128,161);
<pc group of size 170141183460469231731687303715884105728 with 127 generators>
```

The result returned by `PcNormalizedUnitGroupSmallGroup` is equivalent to the following:

Example

```
gap> G := SmallGroup( s, n );
gap> p := PrimePGroup( G );
gap> K := GF( p );
gap> KG := GroupRing( K, G );
gap> PcNormalizedUnitGroup( KG );
```

Nevertheless, `PcNormalizedUnitGroupSmallGroup` is not just a shortcut for such computation. It reads the description of the normalized unit group from the `UnitLib` library and then reconstructs all its necessary attributes and properties. Thus, if you would like to obtain the group algebra KG or the field K and the group G , you should extract them from $V(KG)$, which should be constructed first.

Example

```
gap> V:=PcNormalizedUnitGroup(GroupRing(GF(2),SmallGroup(8,3)));
<pc group of size 128 with 7 generators>
gap> V1:=PcNormalizedUnitGroupSmallGroup(8,3);
<pc group of size 128 with 7 generators>
gap> V1=V;      # two isomorphic groups but not identical objects
false
gap> IdGroup(V)=IdGroup(V1);
true
gap> IsomorphismGroups(V,V1);
[ f1, f2, f3, f4, f5, f6, f7 ] -> [ f1, f2, f3, f4, f5, f6, f7 ]
gap> KG:=UnderlyingGroupRing(V1); # now the correct way
<algebra-with-one over GF(2), with 3 generators>
gap> V1=PcNormalizedUnitGroup(KG); # V1 is an attribute of KG
true
gap> K:=UnderlyingField(KG);
GF(2)
gap> G:=UnderlyingGroup(KG);
<pc group of size 8 with 3 generators>
```

Moreover, the original group G can be embedded into the output of the `PcNormalizedUnitGroupSmallGroup`, as it is shown in the next example:

Example

```

gap> f:=Embedding(G,V1);
[ f1, f2, f3 ] -> [ f1, f2, f4 ]
gap> g:=List(GeneratorsOfGroup(G), x -> x^f );
[ f1, f2, f4 ]
gap> G1:=Subgroup(V1,g);
Group([ f1, f2, f4 ])
gap> IdGroup(G1);
[ 8, 3 ]

```

If the first argument s (the order of a group) is not a power of prime, an error message will appear. If s is greater than 243, you will get a warning telling that the library does not contain $V(KG)$ for G of such order, and you can use only data that you already stored in your `unitlib/userdata` directory with the help of the function `SavePcNormalizedUnitGroup` (2.1.2).

It is worth to mention that for some groups of order 243, the construction of the normalized unit group using `PcNormalizedUnitGroupSmallGroup` already requires some noticeable amount of time. For example, it took about 166 seconds of CPU time to compute `PcNormalizedUnitGroupSmallGroup(243,30)` on Intel Xeon 3.4 GHz with 2048 KB cache.

2.1.2 SavePcNormalizedUnitGroup

- ▷ `SavePcNormalizedUnitGroup(G)` (function)
- ▷ `ParSavePcNormalizedUnitGroup(G)` (function)

Returns: true

Let G be a finite p -group of order s from the Small Groups Library of the GAP system, constructed with the help of `SmallGroup(s,n)` (see `SmallGroup` (**smallgrp: SmallGroup for group order and index**)). Then `SavePcNormalizedUnitGroup(G)` creates the file with the name of the form `us_n.g` in the directory `unitlib/userdata`, and returns true if this file was successfully generated. This file contains the description of the normalized unit group $V(KG)$ of the group algebra of the group G over the field of p elements.

If the order of G is greater than 243, after this you can construct the group $V(KG)$ using `PcNormalizedUnitGroupSmallGroup` (2.1.1) similarly to the previous section. The preliminary warning will be displayed, telling that for such orders you can use only those groups that were already computed by the user and saved to the `unitlib/userdata` directory. If there will be no such file there, you will get an error message, otherwise the computation will begin.

If the order of G is less or equal than 243, then the file will be created in the `unitlib/userdata` directory, but `UnitLib` will continue to use the file with the same name from the appropriate directory in `unitlib/data`. You can compare these two files to make it sure that they are the same.

NOTE THAT:

1. The argument should be the underlying group G and not the normalized unit group $V(KG)$.
2. The argument should be a group from the GAP Small Groups Library constructed with the help of `SmallGroup(s,n)`, otherwise the date consistency may be lost.

`ParSavePcNormalizedUnitGroup` works in the same way, but uses the function `ParPcNormalizedUnitGroup` to parallelise the computation using the GAP package `SCSCP`. Both of the two latter functions are implemented in the file `unitlib/lib/parunits.g`. See [KL10] for the implementation details.

Example

```
gap> SavePcNormalizedUnitGroup( SmallGroup( 256, 56092 ) );
true
gap> PcNormalizedUnitGroupSmallGroup( 256, 56092 );
WARNING : the library of V(KG) for groups of order
256 is not available yet !!!
You can use only groups from the unitlib/userdata directory
in case if you already computed their descriptions
(See the manual for SavePcNormalizedUnitGroup).
#I Description of V(KG) for G=SmallGroup(256,
56092) accepted, started its generation...
<pc group of size
57896044618658097711785492504343953926634992332820282019728792003956564819968
with 255 generators>
```

Chapter 3

Implementation Details

In this chapter we describe the approach used to store the normalized unit group of the group algebra in the library, and to reconstruct the group $V(KG)$ from the stored information.

3.1 Saving the data

To compute the pc-presentation of the normalized unit group of the modular group algebra of a finite p -group we used the function `PcNormalizedUnitGroup` from the **LAGUNA** package. It uses the algorithm described in [Bov98]. See the **LAGUNA** manual [BKRS] for more details.

When this group is computed, the main idea is to use **GAP** function `CodePcGroup` that returns the code for the polycyclic generating sequence of the group, and then to create the group from this code using the **GAP** function `PcGroupCode`.

The resulting code could be very long, and to compress it we used the **GAP** function `HexStringInt` that returns a string that represents the code with hexa-decimal digits (using A-F as digits 10-15). The inverse translation then can be performed with the **GAP** function `IntHexString`. For example, for groups of order 128, this allows to save almost 20 MB of space and reduce the total size of their database to 90 MB.

Furthermore, the library was compressed using the `gzip` program. This allowed us, for example, to reduce the size of the library for groups of order 128 from 90 to 12 MB. Of course, there is some little overhead arising from the uncompression and subsequent translation from hexa-decimal notation, but it is neglectable comparatively with the total time of the computation of $V(KG)$ from scratch.

There is one more thing that needs to be stored together with this code to make it sure that we will correctly identify the underlying group G of the group algebra KG with its image in the pc-presentation of the normalized unit group $V(KG)$.

The group G is extracted from the **GAP** Small Groups Library, so it is always the same, unless its description in the library will be changed (and it will be an important task of **UnitLib** maintaner to update the package in this case!), and here we are safe from inconsistencies.

But the next stage is the computation of generators of the normalized unit group $V(KG)$, and the first step is the dimension basis of the group G , that can be computed using the **LAGUNA** function `DimensionBasis`. To avoid the influence of possible changes in **GAP** or usage of random methods, we store (in compacted form) the information about the dimension basis of G in the **UnitLib**.

All further procedures are implemented inside the **LAGUNA** package, and their result is uniquely determined and predictable.

For the reader interested in more details, the package contains the file `unitlib/lib/genlib.g` with the function `CreatePcNormalizedUnitGroupsLibrary`, that creates library files for groups of a given prime power order.

3.2 Reading the data

To reconstruct the normalized unit group $V(KG)$ from the library, we need only to know the catalogue number of the underlying group G in the GAP Small Groups Library.

We use the same numbering as in the GAP Small Group Library, so UnitLib finds the appropriate library file(s) and reads from it the code for the polycyclic generating sequence of $V(KG)$ and the information about the dimension basis of G used for the computation of this code.

Then $V(KG)$ is created from the code using the GAP function `PcGroupCode` (**Reference: PcGroupCode**). We also create G using the GAP Small Groups Library.

Now to “glue” the group $V(KG)$ with the underlying group G properly, the value of the attribute `DimensionBasis` (**LAGUNA: DimensionBasis**) of G is set in accordance with the data retrieved from the library. This will guarantee the correct construction of `NaturalBijectionToPcNormalizedUnitGroup` (**LAGUNA: NaturalBijectionToPcNormalizedUnitGroup**) and `NaturalBijectionToNormalizedUnitGroup` (**LAGUNA: NaturalBijectionToNormalizedUnitGroup**) by the LAGUNA package.

It remains now to make only several technical steps, such as constructing the group algebra KG over the appropriate field K , and storing KG in the attribute `UnderlyingGroupRing` (**LAGUNA: UnderlyingGroupRing**) of $V(KG)$ and $V(KG)$ in the attribute `PcNormalizedUnitGroup` (**LAGUNA: PcNormalizedUnitGroup**) of KG .

Chapter 4

An example of UnitLib usage

We will finish with several examples of UnitLib usage to give an idea how to work with the package.

In the first example we retrieve from the library the normalized unit group of the group algebra of the dihedral group of order 128 over the field of two elements, compute its center and find some of its properties, and then check that the group generated by these generators expressed in terms of group algebra elements is abelian:

Example

```
gap> IdGroup(DihedralGroup(128));
[ 128, 161 ]
gap> V := PcNormalizedUnitGroupSmallGroup( 128, 161 );
<pc group of size 170141183460469231731687303715884105728 with 127 generators>
gap> C := Center( V );
<pc group with 34 generators>
gap> gens := MinimalGeneratingSet( C );
gap> Length(gens);
19
gap> Size(C);
17179869184
gap> AbelianInvariants(C);
[ 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 4, 4, 4, 4, 8, 8, 16, 32 ]
gap> KG := UnderlyingGroupRing( V );
<algebra-with-one over GF(2), with 7 generators>
gap> f := NaturalBijectionToNormalizedUnitGroup( KG );
gap> IsAbelian( Group( List( gens, x -> x^f ) ) );
true
```

In the second example we will check the conjecture about the coincidence of the lower and upper Lie nilpotency indices of the modular group algebras for all non-abelian groups of order 64.

It is known that these indices coincide for p -groups with $p > 3$ [BP92], but in the general case the problem remains open.

The indices $t_L(G)$ and $t^L(G)$ can be computed using the LAGUNA package. While the upper Lie nilpotency index can be expressed only in terms of the underlying group G , the lower Lie nilpotency index is determined by the formula $t_L(G) = \text{cl } V(KG) + 1$ [Du92], and can be computed immediately whenever $V(KG)$ is known.

In the program below we enumerate all groups of size 64 and check the conjecture (we do not exclude from consideration some particular cases when the conjecture is known to be true for $p = 2$, because this is beyond the task of this manual).

Example

```
gap> for n in [ 1 .. NrSmallGroups( 64 ) ] do
> if not IsAbelian( SmallGroup( 64, n ) ) then
>   V := PcNormalizedUnitGroupSmallGroup( 64, n );
>   KG := UnderlyingGroupRing( V );
>   if LieLowerNilpotencyIndex( KG ) <>
>     LieUpperNilpotencyIndex( KG ) then
>     Print( n, " - counterexample !!! \n" );
>     break;
>   fi;
> fi;
> od;
```

Thus, the test was finished without finding a counterexample.

In the next example we will answer the question about possible nilpotency classes of normalized unit groups of modular group algebras of nonabelian groups of order 64:

Example

```
gap> cl := [];
gap> for n in [ 1 .. NrSmallGroups( 64 ) ] do
> if not IsAbelian( SmallGroup( 64, n ) ) then
>   V := PcNormalizedUnitGroupSmallGroup( 64, n );
>   AddSet( cl, NilpotencyClassOfGroup( V ) );
> fi;
> od;
gap> cl;
[ 2, 3, 4, 5, 6, 7, 8, 16 ]
```

With UnitLib you can perform the computation from the last example in several hours on a modern computer. Without UnitLib you will spend the same time to compute only several normalized unit groups $V(KG)$ for groups of order 128 with the help of the LAGUNA package. Note that without LAGUNA such computation would not be feasible at all.

References

- [BKRS] V. Bovdi, O. Konovalov, R. Rossmanith, and C. Schneider. LAGUNA — Lie AlGebras and UNits of group Algebras. GAP4 package (<https://gap-packages.github.io/laguna/>). 4, 6, 10
- [Bov98] Adalbert Bovdi. Generators of the units of the modular group algebra of a finite p -group. In *Methods in ring theory (Levico Terme, 1997)*, volume 198 of *Lecture Notes in Pure and Appl. Math.*, pages 49–62. Dekker, New York, 1998. 10
- [BP92] Ashwani K. Bhandari and I. B. S. Passi. Lie-nilpotency indices of group algebras. *Bull. London Math. Soc.*, 24(1):68–70, 1992. 12
- [Du92] Xian Kun Du. The centers of a radical ring. *Canad. Math. Bull.*, 35(2):174–179, 1992. 12
- [KL10] O. Konovalov and S. Linton. Parallel computations in modular group algebras. In *Proceedings of the 4th International Workshop on Parallel and Symbolic Computation*, PASCO '10, pages 141–149. Association for Computing Machinery, 2010. 8

Index

ParSavePcNormalizedUnitGroup, [8](#)
PcNormalizedUnitGroupSmallGroup, [6](#)
SavePcNormalizedUnitGroup, [8](#)
UnitLib package, [2](#)